

# GRAPHES EN 3D

## (1ère partie)

Bernard Dupont

[Bernard.Dupont@univ-lille1.fr](mailto:Bernard.Dupont@univ-lille1.fr)

Ce chapitre est consacré à la représentation en 3 dimensions. Le sujet est vaste et il est traité en deux parties pour des raisons de commodité.

La première partie s'attaque à la représentation graphique des fonctions de deux variables réelles, donc aux surfaces d'équation  $z=f(x, y)$  où  $x$  est l'abscisse d'un point,  $y$  est son ordonnée et  $z$  sa côte ou altitude. La commande de base est `plot3d` dont les règles basiques d'utilisation sont exposées dans une première section. De nombreuses options lui sont associées afin d'améliorer le graphe et elles font l'objet de la deuxième section. En pratique, l'utilisateur fera également appel aux ressources du paquetage `plots` sur lesquelles on ne reviendra pas ici puisqu'elles ont été exposées dans le chapitre consacré aux graphes en deux dimensions.


### ▼ Tracés élémentaires

Soit l'équation  $z=f(x, y)$  où  $f$  est explicitée. On veut tracer la surface représentative de  $f$  sur le sous-ensemble  $[a; b] \times [c; d]$  de  $\mathbb{R}^2$  ( $a < b$  et  $c < d$ ). La fonction mathématique  $f$  a été définie comme une expression ou comme fonction-procédure. Dans tous les cas, on invoque la commande jumelle de `plot` (réservée aux graphes en deux dimensions), soit `plot3d`.

### ▼ Tracé d'une expression

Si `xp` est une expression des variables libres  $x$  et  $y$ , la commande `plot3d(xp, x=a..b, y=c..d)` renvoie un graphe donnant la surface sur  $[a; b] \times [c; d]$ .

```
> restart;  
xp:=cos(x^2)+cos(y);plot3d(xp,x=1..3,y=-3..3);  
xp := cos(x2) + cos(y)
```




En cliquant gauche sur le graphique, le pointeur prend la forme d'une flèche courbée et on constate que la cartouche Plot se met automatiquement en surbrillance, ce qui signale que la

barre d'outils spécifique aux graphiques en 3D - une barre contextuelle - apparaît sous la barre de menu.

Elle donne la possibilité de tracer les axes, ou de placer la surface dans une boîte, ou de représenter la surface avec ou sans les lignes de niveau, etc.. A explorer.


Notons qu'il est possible de modifier la position de l'observateur en sélectionnant le pointeur sur le mode rotate, en cliquant "gauche" sur la surface puis en "déplaçant" la souris sans relâcher la pression.





Le nouveau pointeur permet de modifier les angles de vue de la surface. On peut donc la torturer dans tous les sens par déplacement de la souris sans relâcher la pression jusqu'à ce qu'on trouve le point de vue le plus intéressant. Les angles de vue sont reportés dans la barre contextuelle. Par défaut, on lit



Les propriétés du pointeur sont paramétrables par le bouton de droite :



L'option  permet de zoomer ou réduire la surface alors que l'option  la déplace à l'intérieur du cadre.

Le style de la surface est défini au moyen du bouton à options :



Il ne faut pas hésiter à explorer les différents styles proposés (surfaces lisses, courbes de niveau, maillages, etc.), sur lesquels on reviendra en détail plus loin.

Par défaut, Maple ne trace pas d'axes, ce qui nuit grandement à la lisibilité de la surface. L'utilisateur choisit son système d'axes par le bouton à options (qui seront commentées plus loin) :




Enfin, le bouton **1:1** fait la bascule entre le repère choisi par défaut par le logiciel et la représentation en repère orthonormé.

Toutes ces options contribuent grandement à améliorer la présentation de la surface obtenue par défaut avec **plot3d**. Cela dit, il faut bien comprendre que leur effet n'est que temporaire : une nouvelle validation de l'instruction renvoie en effet le graphique primitif. Pour fixer un rendu particulier, il faut le programmer, c'est à dire insérer explicitement des options dans la commande. La section 2 sera entièrement consacrée aux options de **plot3d**.

## ▼ Tracé d'une fonction-procédure

Si **fonc** est une fonction Maple de deux variables, la commande **plot3d(fonc, a..b, c..d)** renvoie un graphe donnant la surface sur  $[a;b] \times [c;d]$ . On utilise alors la barre des outils spécifiques aux graphiques comme dans le paragraphe précédent.

```
> f := (x, y) -> cos(x^2) + cos(y); plot3d(f, 1..3, -3..3);
      f := (x, y) -> cos(x2) + cos(y)
```






On peut encore accéder au menu contextuel en sélectionnant le graphique.

## Tracé simultané de plusieurs surfaces

On trace plusieurs surfaces dans un même graphique par la commande `plot3d({f1,f2,..,fn},x=a..b,y=c..d)` si les `fi` sont des expressions ou `plot3d({f1,f2,..,fn},a..b,c..d)` si les `fi` sont des fonctions. Les ensembles d'expressions ou de fonctions peuvent être remplacés par des listes.

```
> f1:=(x,y)->(x^2+y^2)^(1/2):#fonction générant un cône
plot3d(f1,-1..1,-1..1);#représentation graphique du cône
f2:=(x,y)->(x^2-y^2)/(x^2+y^2):#fonction générant le
conoïde de Plücker
plot3d(f2,-1..1,-1..1);#représentation graphique du
conoïde
plot3d({f1,f2},-1..1,-1..1);#représentation simultanée du
cône et du conoïde
```





## Représentation d'un point dans l'espace

Soit le point  $M$  de coordonnées  $(a, b, c)$ . Sa représentation dans l'espace est donnée par `plot3d([a,b,c],x=xm..xM,y=ym..yM,style=point)` où les options d'intervalles pour  $x$  et  $y$  doivent figurer obligatoirement.

```
> plot3d([1,1,1],x=0..2,y=0..2,style=point);
```



.

└ On aura intérêt à utiliser les options d'axes pour améliorer la représentation.


## ▼ **Assignment d'un graphique**

L'assignation d'un graphique se fait suivant les mêmes règles que celles données pour les graphes en deux dimensions. Quand on attribue un nom à un graphique, il faut terminer l'instruction par "deux points" et non "point virgule", faute de quoi l'écho est une succession de coordonnées et d'indications optionnelles réunies dans un objet *PLOT3D(...)*. On obtient le graphique en tapant son nom clôturé par un "point virgule".

```
> graf1:=plot3d(f1,-1..1,-1..1);graf1;  
graf2:=plot3d(f2,-1..1,-1..1):graf2;  
graf3:=plot3d({f1,f2},-1..1,-1..1):graf3;  
graf1 := PLOT3D(...)
```








L'assignation a cet avantage qu'on peut exploiter les ressources de la commande **display** du paquetage **plots**.

```
> with(plots):  
  display({graf1,graf2},title="Réunion d'un cône et d'un  
  conoïde de Plucker",titlefont=[TimesNewRoman,14,Bold]);
```

Réunion d'un cône et d'un conoïde de Plucker



## Options de plot3d

Les options de **plot3d** sont nombreuses. On en a le détail par la commande :

```
> ?plot3d,options;
```

qui renvoie directement dans l'aide en ligne. On n'évoque ici que les plus courantes. Rappelons encore une fois que la plupart des options sont disponibles en sélectionnant le graphe par un clic gauche puis en explorant la barre d'outils spécifique aux graphes ou en faisant un clic droit sur la surface ou encore en sélectionnant une commande dans le menu "plot" qui apparaît dans la barre

de menus.

## ▼ Axes (**axes, tickmarks, labels**)

Par défaut, la commande `plot3d` ne fait pas apparaître les axes des abscisses, ordonnées et altitudes alors qu'on a souvent besoin d'informations sur les coordonnées. L'option **axes** accepte les valeurs suivantes :

**BOXED** ou **boxed**, pour faire apparaître le graphique dans un cube gradué,

**NORMAL** ou **normal**, qui correspond au tracé usuel d'une surface dans un espace muni d'un repère orthogonal et d'une origine où tous les axes passent,

**FRAME** ou **frame**, qui correspond au tracé usuel d'une surface dans un espace muni d'un repère orthogonal sans origine fixe,

**NONE**, qui est la valeur par défaut.

Les axes sont gradués à la demande avec l'option **tickmarks=[abs,ord,alt]** où **abs** fixe le nombre de graduations sur l'axe des abscisses, **ord** le nombre de graduations sur l'axe des ordonnées et **alt** le nombre de graduations sur l'axe des altitudes. Alternativement, on peut forcer le logiciel à adopter son choix de graduations en donnant des listes de coordonnées. L'option **axesfont** règle la typographie des graduations dans les limites fixées par **fonts** (voir le chapitre sur les graphes en deux dimensions)



Les axes se labellent avec l'option **labels=[x,y,z]** où **x**, **y** et **z** sont des chaînes de caractères placées entre guillemets ou écrites avec **typeset**. Par défaut, Maple retient les variables présentes dans l'expression. L'option **labeldirections=[orx,ory,orz]** spécifie l'orientation des labels; les valeurs de **orx**, **ory** et **orz** peuvent être **horizontal** (qui est la valeur par défaut) ou **vertical**. L'option **labelfont=[police,style,taille]** sert à choisir la police de caractères, son style et sa taille pour les labels; on doit retenir l'une des polices disponibles dans l'option générale **font**.

## ▼ Réglage du repère orthogonal (**scaling**)

Par défaut, Maple choisit une base de vecteurs orthogonaux dont les longueurs permettent de remplir au mieux une page de format A4. Des effets de perspective fâcheux ou indésirables peuvent en découler. On peut forcer le logiciel à adopter une base orthonormale en fixant l'option **scaling** à **constrained** (par défaut, l'option est à **unconstrained**).

```
> restart;  
plot3d(cos(x^2)+cos(y),x=1..3,y=-3..3,axes=boxed);  
plot3d(cos(x^2)+cos(y),x=1..3,y=-3..3,scaling=  
constrained,axes=boxed);
```







### ▼ Ajustement du tracé (**view**)



Par défaut, Maple représente tous les points de coordonnées  $(x, y, f(x, y))$  où  $(x, y)$  appartient au sous-ensemble  $[a, b] \times [c, d]$ . On limite les valeurs prises par l'image  $z = f(x, y)$  par l'option **view=zmin..zmax**.

```
> restart; f := (x, y) -> x^2 + y^2; plot3d(f, -4..4, -4..4, axes=
normal); plot3d(f, -4..4, -4..4, view=2..12, axes=normal);
f := (x, y) -> x^2 + y^2
```



Cette option est particulièrement utile en présence d'une limite infinie :

```
> restart;plot3d(1/(x^2+y^2),x=-2..2,y=-2..2,axes=frame);
plot3d(1/(x^2+y^2),x=-2..2,y=-2..2,view=0..5,axes=frame);
```



### ▼ Point de vue (**projection, orientation**)

La lisibilité d'une surface dépend beaucoup du point de vue du spectateur. Avec Maple, ce dernier peut virtuellement se rapprocher de la surface ou s'en éloigner, et tourner tout autour pour choisir le meilleur angle de vue.




L'option **projection=r** où **r** est un réel compris entre 0 et 1 définit la perspective. Si **r=0**, tout se passe comme si le spectateur a le nez collé sur la surface; on peut aussi écrire **projection=FISHEYE**. Si **r=1**, c'est le contraire : le spectateur est infiniment éloigné et regarde la surface avec des jumelles; on peut aussi écrire **projection=ORTHOGONAL**, qui est la valeur par défaut. Toutes les situations intermédiaires sont possibles; quand **r=0.5**, on peut aussi écrire **projection=NORMAL**. L'exemple suivant montre la même surface vue selon une perspective différente.

```
> restart; plot3d(sin(x^2)+cos(y), x=-2..2, y=-5..5,
  projection=FISHEYE, axes=boxed);
```

```

plot3d(sin(x^2)+cos(y),x=-2..2,y=-5..5,projection=NORMAL,
axes=boxed);
plot3d(sin(x^2)+cos(y),x=-2..2,y=-5..5,projection=
ORTHOgonal,axes=boxed);



```




L'option **orientation=[theta,phi]** est sans conteste l'une des plus importantes. C'est d'elle dont dépend largement la qualité de présentation de la surface représentée puisqu'elle fixe l'angle de vue. Par défaut, Maple situe l'oeil de l'observateur en un point situé à 45 degrés de latitude et 45 degrés de longitude; mais cette position peut ne pas être la meilleure pour appréhender les particularités d'une surface et on peut régler la longitude, à savoir l'angle

theta ( $\theta$ ) exprimé en degrés, et la latitude, à savoir l'angle phi ( $\varphi$ ) exprimé lui aussi en degrés. D'un point de vue mathématique, ces 2 angles sont les coordonnées sphériques de l'oeil. Dans la pratique, on détermine les "bons" angles en tâtonnant à partir des facilités offertes par le pointeur de la souris et le menu contextuel. Un clic gauche sur la figure permet d'effectuer toutes les rotations possibles et imaginables dont les angles  $\theta$  et  $\varphi$  correspondants sont mesurés "en temps réel" dans la partie gauche du menu contextuel. Il suffit de lire les angles caractérisant le meilleur rendu et de les reporter comme paramètres de l'option.

```
> restart;plot3d(x^2-y^2,x=-1..1,y=-1..1,axes=boxed,
orientation=[63,65]);#forme quadratique générant un col
plot3d(x^2-y^2,x=-1..1,y=-1..1,axes=boxed,orientation=[0,
0]);#vue par dessus
plot3d(x^2-y^2,x=-1..1,y=-1..1,axes=boxed,orientation=
[45,90]);#vue en se plaçant sur le plan équatorial
```







### ▼ Précision du tracé (**numpoints**, **grid**)

Par défaut, Maple trace une surface en reliant 625 points (soit 25x25). On augmente la précision du graphique par l'option **numpoints=N**, **N** étant un entier naturel.

```
> restart;
plot3d(x^(1/3)*y^(2/3),x=0..10,y=0..10,axes=normal,
orientation=[26,85]);#représentation par défaut d'une
Cobb-Douglas
plot3d(x^(1/3)*y^(2/3),x=0..10,y=0..10,axes=normal,
orientation=[26,85],numpoints=2000);#représentation plus
"fine"
```









Quant à elle, l'option **grid=[m,n]** sert à fixer les dimensions de la grille rectangulaire sur laquelle se trouvent régulièrement espacés les abscisses et les ordonnées.

```
> restart;plot3d(x^(1/3)*y^(2/3),x=0..10,y=0..10,axes=
normal,orientation=[26,85],grid=[5,30]);
```



L'option **gridstyle** spécifie le type de maillage désiré pour la grille, qui est soit **rectangular** (par défaut), soit **triangular**.

```
> plot3d(x^(1/3)*y^(2/3),x=0..10,y=0..10,axes=normal,
orientation=[26,85],gridstyle=triangular);
```



## ▼ Style de tracé (**style**)



L'option **style** indique comment la surface doit être représentée. Par défaut, Maple utilise **style=patch** (équivalent à l'alias **surfacewireframe**), qui donne une surface colorée avec le maillage des points. Selon les besoins, on peut fixer le style à **point** (la surface est rendue par des points); **hidden** (équivalent à l'alias **wireframeopaque**; la surface est rendue par le maillage des points et seuls les points visibles par l'observateur sont représentés); **wireframe** (équivalent à l'alias **line**; tous les points, même ceux qui devraient rester invisibles à l'observateur, sont reliés par un maillage); **contour** (met en évidence les courbes de niveau); **patchnogrid** (équivalent à **surface**; la surface est colorée mais le maillage des points est supprimé); **patchcontour** (équivalent à **surfacecontour**; la surface est colorée et met en évidence les lignes de niveau).


Pour certaines options sont associées des "sous" options. En particulier :

- Quand la surface fait apparaître les points, on peut choisir le symbole qui les représentent par **symbol=point** ou **box** (carré) ou **circle** (cercle) ou **cross** (croix) ou **diamond** (triangle). La taille du symbole est fixée par l'option **symbolsize=N**, où **N** est un entier compris entre 0 et 10.
- Quand la surface fait apparaître des lignes de niveau, on peut fixer leur style par **linestyle=solid** (ligne continue, qui est l'option par défaut) ou **dot** (ligne pointillée) ou **dash** (succession de tirets) ou **dashdot** (succession de points et de tirets). L'épaisseur de la ligne est fixée par l'option **thickness=N**, où **N** est un entier naturel (égal à 0 par défaut).

```
> restart;
plot3d(x^2+y^2+5*x*y,x=-4..4,y=-4..4,orientation=[126,
58],axes=boxed);#style par défaut
plot3d(x^2+y^2+5*x*y,x=-4..4,y=-4..4,style=patchnogrid,
orientation=[126,58],axes=boxed);#style fixé à surface
plot3d(x^2+y^2+5*x*y,x=-4..4,y=-4..4,style=patchcontour,
linestyle=dot,contours=20,orientation=[126,58],axes=
boxed);#Les lignes de niveau sont en pointillés. Leur
```

nombre est fixé par l'option contours=N, N valant par défaut 10.






### ▼ Colorisation (**color** ou **colour**)

Par défaut, Maple colorise la surface. Généralement, le rendu est satisfaisant, mais dans certains cas il peut être utile de contrôler le nuancier.

L'option **color**, qui peut aussi s'écrire **colour**, offre de très nombreuses possibilités et on aura intérêt à se référer à l'aide en ligne pour les explorer toutes. Une première possibilité est d'utiliser **color=c**, où **c** est une couleur prédéfinie, dont on a donné la liste dans le chapitre consacré aux graphes en deux dimensions. La surface est alors unicolore. La seconde est de nuancer la surface à l'aide d'un générateur de couleurs qui peut être une fonction des trois coordonnées des points représentés, ou de leur abscisse et de leur ordonnée, ou de la seule côte. Sauf exception rarissime, mieux vaut s'en tenir à la colorisation par défaut ou une colorisation uniforme.

```
> restart;plot3d(sin(x^2)+cos(y),x=-2..2,y=-5..5,axes=
  boxed,color=cyan);
```



### ▼ Modelé (**light**, **lightmodel** et **glossiness**)

Pour les experts pointilleux du rendu parfait d'une surface, l'option **light**, de syntaxe **light=[phi,theta,r,v,b]**, est précieuse. Elle simule l'effet d'un projecteur positionné à  $\phi$  radians de latitude et  $\theta$  radians de longitude délivrant une lumière dont le coloris est choisi dans le modèle de couleurs RVB.

Les moins exigeants se contenteront de puiser dans un modelé pré-programmé par le biais de l'option alternative **lightmodel=lm** où **lm** peut prendre les valeurs **light1**, **light 2**, **light 3** ou **light4**.

Comble du professionnalisme : on peut encore régler la luminance de la surface par l'option **glossiness=g** où **g** est un réel compris entre 0 (pas de réflexion) et 1 (réflexion maximale).



### ▼ Ombrage (**shading**)

L'option **shading** spécifie comment la surface doit être coloriée. Elle admet les valeurs **xyz** ou **xy** ou **z** ou **zgrayscale** ou **zhue** ou **none** (absence de coloriage). Si une couleur particulière a été choisie par l'option **color**, l'option **shading** n'a aucun effet.


```
> plot3d(sin(x^2)+cos(y),x=-2..2,y=-5..5,axes=boxed,
shading=none);
plot3d(sin(x^2)+cos(y),x=-2..2,y=-5..5,axes=boxed,
```

```
shading=zgrayscale);#l'altitude définit la luminance du  
gris : gris foncé pour les points bas; gris clairs pour  
les points élevés
```

```
plot3d(sin(x^2)+cos(y),x=-2..2,y=-5..5,axes=boxed,  
shading=zhue);#l'altitude définit la couleur : les  
couleurs froides pour les points bas; les couleurs  
chaudes pour les points élevés
```








### ▼ Tranparence (**transparency**)

Par défaut, Maple cache les parties de la surface masquées par les portions situées au premier plan. Si on veut voir l'entièreté de la surface, il faut régler l'intensité de la transparence avec l'option **transparency=r**, où **r** est un réel compris entre 0 (valeur par défaut équivalant à l'absence de transparence) et 1 (totale transparence).

```
> restart; plot3d(x^2-y^2, x=-1..1, y=-1..1, axes=boxed,
  transparency=0.5, orientation=[77,59]);
```



### ▼ Titre (**title, caption**)

Les commandes **title="texte"** et **caption="texte"** permettent d'insérer un titre, de facture assez grossière au demeurant, respectivement au dessus et en dessous du graphique. Leur argument est une chaîne de caractères, donc les guillemets sont obligatoires. L'option supplémentaire **titlefont=[police, style, taille]** précise le choix de la fonte du titre, à condition d'appartenir aux fontes tolérées par l'option **font**. La fonte du titre-bandeau

est immédiatement précisée dans une liste composée du texte puis de **font**. Un titre peut contenir plusieurs lignes; le passage à la ligne suivante se fait par l'insertion du code `\n` à la fin d'une ligne.

```
> restart;plot3d(x^2-y^2,x=-1..1,y=-1..1,axes=boxed,
transparency=0.5,orientation=[77,59],title="Forme
quadratique \n générant un col",titlefont=[courrier,bold,
12],caption=["Figure1.",font=[TimesNewRoman,14,Bold]]);
```

## Forme quadratique générant un col




Figure1.


Il est souvent nécessaire de taper du texte en mode mathématique en sus d'une partie en mode texte dans le titre, par exemple pour écrire "Figure 1. Représentation graphique de la fonction  $f(x, y) = x^2 - y^2$ ". Malheureusement, il n'est pas possible d'obtenir ce résultat dans la version 12 de Maple. Cette possibilité n'est réservée qu'aux graphes en deux dimensions.

### ▼ Insertion de texte dans un graphique (**textplot3d**)

Pour documenter un graphique en 3D, il faut utiliser la commande **textplot3d** qui appartient au paquetage **plots**. Plus précisément, deux graphiques doivent être construits : le graphique proprement dit avec **plot3d**; les textes à insérer avec **textplot3d**. Ils sont ensuite réunis avec la commande **display**.

La syntaxe générale de la commande est **textplot3d([abs,ord,alt,"texte à insérer"],options)** où le texte à insérer est une chaîne de caractères centrée sur un point de coordonnées **abs**, **ord** et **alt**. Par défaut, le milieu de la chaîne de caractères est centré sur ce point, mais on peut modifier cette présentation en ajoutant l'option **align=ABOVE** (au dessus du point de référence) ou **BELOW** (en dessous) ou **LEFT** (aligné à gauche sur ce point) ou **RIGHT** (aligné à droite sur ce point). On peut en outre ajouter n'importe quelle autre option de **plot3d**.

```
> restart;with(plots):
graf1:=plot3d(x^2-y^2,x=-1..1,y=-1..1,axes=boxed,
transparency=0.5,orientation=[77,59]):
graf2:=textplot3d([0,0,0.2,"COL"],color=red):
display3d(graf1,graf2);
```



Comme pour les titres, il n'est pas possible d'écrire du texte en mode mathématique ou d'utiliser les symboles des palettes, ce qui restreint sensiblement l'intérêt de cette commande.